
９．サブルーチンとマクロ定義

サブルーチン（Sub Routine）とはメインプログラム（メインルーチン）とは別にブロ

ック化された一連の処理を行うプログラムのことをいう。サブルーチン化するメリットの

１つは同じ処理を行うプログラムを何度も記述する必要がなくなること，もう１つはメイ

ンプログラムをサブルーチンによってブロッ

ク化することにより，プログラムの修正やサ

ブルーチンの入れ替え（結合状態）が容易に

なることである。

サブルーチンによるプログラムの流れは右

図のようであり，サブルーチンへの移行は

「CALL」命令で行われ，その終了（メインル

ーチンへの戻り）は「RET」命令で行われる。

ＣＰＵが行う命令が格納されたメモリのア

ドレスは「CS（コードセグメント）」と「IP

（インストラクションポインタ）」で決定さ

れる。「CALL」命令が実行されるとＣＰ

Ｕはサブルーチンから戻った時に実行す

る命令のアドレス CSと IP（CALL命令

の次の命令）を「SS（スタックセグメン

ト）」と「SP（スタックポインタ）」で

決まるメモリの物理アドレスへ退避（保

存）し，サブルーチンのアドレスを CS

と IP にセットする。SP の値は退避され

るごとに２づつ減じられる。一方，サブ

ルーチンで「RET」命令が実行されると，

先の退避した戻りアドレスをスタックか

ら CSと IPに復帰（ロード）し，次の命

令へ移る。なお，サブルーチンがメインルーチンと同一セグメント内に記述される場合（通

常はこの記述である）にはスタックには「IP」だけが退避，復帰される。プログラムの実

行状態はこのようであるが，ユーザーは特に「SP」の値を考慮しなくてよい（「SS」と「SP」

の値はＤＯＳが実行ファイルをローディングする時に決める）。

サブルーチンプログラムを作成する場合には，通常メインプログラムとは別のソースフ

ァイルで作成し，それぞれをコンパイルして「LINK」で結合するか，あるいはメインソー

スプログラム中で「INCLUDE」ディレクティブと呼ばれる関数によって結合して実行フ

ァイルを作成する。両者の方法によって簡単なプログラムを作成してみよう。

０から９までの数字を２列に縦に並べて表示するプログラムを作成するが，まず，

「INCLUDE」ディレクティブによって結合する場合を説明する。

「EX91.C」「EX91A.ASM」はサブルーチンを使用しないプログラムである。ＡＳＭプ

ログラムにおいて，同じような処理を行うプログラムを２度記述している。これをサブル

ーチン化し，メインプログラムと同一ソースファイル中に記述したＡＳＭプログラムが

「EX91A1.ASM」である。サブルーチン名は「DSUB」としたが，予約変数名以外であれ

ば何でもよい。また，大文字でも小文字でもよ

い（コンパイルでは判別しない）。サブルーチ

ンは「proc near」で１つのプロシジャーとし

て定義し，「endp」でその終りを示す。

「EX91A2.ASM」と「EX91AD.INC」はサ

ブルーチン「DSUB」を別のファイルとして作

成し，「INCLUDE」ディレクティブで結合す

/* EX91.C */

main()

{

disp1();

}

;EX91A.ASM

_TEXT segment byte public 'CODE'

assume  cs:_TEXT

public  _disp1

_disp1 proc near

MOV  AX,0A000H

MOV  ES,AX

MOV  AL,30H

MOV  BX,40

MOV  CX,10

LP1: MOV  ES:[BX],AL

INC  AL

ADD  BX,00A0H

LOOP LP1

MOV  AL,30H

MOV  BX,60

MOV  CX,10

LP2: MOV  ES:[BX],AL

INC  AL

ADD  BX,00A0H

LOOP LP2

RET

_disp1 endp

_TEXT ends

end

;EX91A1.ASM

_TEXT segment byte public 'CODE'

assume  cs:_TEXT

public  _disp1

_disp1  proc near

MOV  AX,0A000H

MOV  ES,AX

MOV  AL,30H

MOV  BX,40

CALL DSUB

MOV  AL,30H

MOV  BX,60

CALL DSUB

RET

_disp1 endp

DSUB proc near

MOV  CX,10

LP: MOV  ES:[BX],AL

INC  AL

ADD  BX,00A0H

LOOP LP

RET

DSUB endp

_TEXT ends

end

;EX91A2.ASM

_TEXT segment byte public 'CODE'

assume  cs:_TEXT

public  _disp1

_disp1  proc near

MOV  AX,0A000H

MOV  ES,AX

MOV  AL,30H

MOV  BX,40

CALL DSUB

MOV  AL,30H

MOV  BX,60

CALL DSUB

RET

_disp1 endp

INCLUDE B:¥INCLUDE¥EX91AD.INC

_TEXT ends

end

;B:¥INCLUDE¥EX91AD.INC

DSUB proc near

MOV  CX,10

LP: MOV  ES:[BX],AL

INC  AL

ADD  BX,00A0H

LOOP LP

RET

DSUB endp

る場合のプログラムである。サブルーチン「DSUB」の部分をそのまま別ファイルとし，

メインソースプログラム内の同一セグメント内で「INCLUDE」を定義する。なお，ここ

ではインクルードされるファイル「EX91AD.INC」（拡張子は予約名以外であれば何でも

よい）をドライブＢのサブディレクトリ「INCLUDE」内に作成した場合である。ＡＳＭ

ファイルのコンパイルでは「EX91A2.ASM」だけ指定することにより，インクルードファ

イルを結合してコンパイルを行い，１つのオブジェクトファイル「EX91A2.OBJ」を作成

する。この「INCLUDE」ディレクティブで結合する方法は，単に１つのソースプログラ

ムを２つに分けただけであるので，このインクルードされるファイルはそれ自身で独立し

たソースファイルではない。すなわち，インクルードファイルだけをコンパイルすること

はできない。

では，次に，サブルーチンも独立したソースファイルで作成し，「LINK」よって結合

する場合のプログラムを「EX91A3.ASM」「EX91AS.ASM」に示す。この時にはサブル

ーチンのソースプログラムもまたＡＳＭファイル（拡張子が「.ASM」）である。メインソ

ースファイルではサブルーチン「DSUB」が別のソースファイルで定義されていることを

指示するために「EXTRN（External ディレクティブ）」宣言を行わなければならない。

サブルーチン名に続く「:near」は同一セグメント内にサブルーチンを定義することを意味

し，サブルーチンソースファイル「EX91AS.ASM」内の「DSUB proc near」に対応する。

「EX91AS.ASM」でも，それ自身がＣ言語に対応した形式記述でなければならないので，

メインソースプログラムと同様な記述となる。サブルーチン「DSUB」は別のソースファ

イル（ここでは「EX91A3.ASM」）で参照（呼び出される）されるので，「Public」宣言

でその許可を行う。

２つのプログラムはそれぞれ独立してコンパイルを行い，リンカーによって結合される。

「extrn」と「public」はリンク時にこの２つのオブジェクトファイルを結合するための情

報となる。

;EX91A3.ASM

_TEXT segment byte public 'CODE'

assume  cs:_TEXT

public  _disp1

extrn DSUB:near

_disp1  proc near

MOV  AX,0A000H

MOV  ES,AX

MOV  AL,30H

MOV  BX,40

CALL DSUB

MOV  AL,30H

MOV  BX,60

CALL DSUB

RET

_disp1 endp

_TEXT ends

end

;EX91AS.ASM

_TEXT segment byte public 'CODE'

assume  cs:_TEXT

public  DSUB

DSUB proc near

MOV  CX,10

LP: MOV  ES:[BX],AL

INC  AL

ADD  BX,00A0H

LOOP LP

RET

DSUB endp

_TEXT ends

end

「INCLUDE」ディレクティブで結合する方法に比べて，独立したソースファイルを作

成しリンカーで結合する方法では，それぞれを別々にコンパイルするので，プログラム記

述などのエラーが発生した場合などの処置が容易となる。２つの方法におけるコンパイル

とリンクの流れは下図のようである。なお，現在作成している制御プログラムは全て

「INCLUDE」ディレクティブで結合する方法としている。

［マクロ定義］

マクロ定義（Macro Define）とはマクロアセンブラ

で用意されているディレクティブであり，１連のプロ

グラムの記述形式を定義するものである。先の

「EX91A.ASM」をマクロ定義を用いて，記述すると

「EX92A.ASM」のようになる。

記述定義は

マクロ名 MACRO 引数,引数,引数,・・・

ENDM

であり，ソースプログラムの最初のセグメント外で行

う。「マクロ名」および「引数」は予約変数以外であ

る。「引数」は数値，変数および文字であるが，文字

を引き渡す場合には，例えば

MOV AX,OFFSET &D1

のように先頭に「＆」を付ける。「LOCAL」はマクロ

定義中で使用するラベルを定義を行う。

一方，プログラム中では

マクロ名 引数,引数,引数,・・・

;EX92A.ASM

DSUB MACRO D1,D2

LOCAL LP

MOV  AL,D1

MOV  BX,D2

MOV  CX,10

LP: MOV  ES:[BX],AL

INC  AL

ADD  BX,00A0H

LOOP LP

ENDM

_TEXT segment byte public 'CODE'

assume  cs:_TEXT

public  _disp1

_disp1 proc near

MOV  AX,0A000H

MOV  ES,AX

DSUB 30H,40

DSUB 30H,60

RET

_disp1 endp

_TEXT ends

end

と記述し，コンパイルによってそれぞれの引数がマクロ定義の変数に渡され，プログラム

中に展開される。なお，引数は「，」で区切り，ブランクを入れてはいけない。

このマクロ定義は見かけ上，先のサブルーチンのようであるが，コンパイルされた結果

のファイルは「EX91A.ASM」と全く同じものとなる。したがって，サブルーチン化した

プログラムに比べて使用するプログラムメモリは大きくなるが，「CALL」や「RET」に

伴う処理を行わないために，プログラムの実行時間は速くなる。

マクロアセンブラで使用するものとして，「EQU」ディレクティブがある。これは数値

を変数で定義するものであり，次の形式でソースプログラムの最初（マクロ定義より先に）

に行う。

名前　EQU　データ

ソースプログラム中では全て名前（予約名以外の英字で始まるもの）で記述できる。プロ

グラム中で数値をそのまま記述する代わりに「EQU」ディレクティブで定義しておけば，

簡単にデータを変えることができる。

マクロアセンブラではその他にも多くのディレクティブがあるが，それらについてはマ

ニュアルを参照されたい。


